

## The EBNF specification of the 3APL language for individual agents

```

<Program> ::=      "Program" <ident>
                  ( "Load" <ident> )?
                  "Capabilities:" ( <capabilities> )?
                  "BeliefBase:" ( <beliefs> )?
                  "GoalBase:" ( <goals> )?
                  "PlanBase:" ( <plans> )?
                  "PG - rules:" ( <p_rules> )?
                  "PR - rules:" ( <r_rules> )?

<capabilities> ::= <capability> ( "," <capability> )*
<capability> ::=  "{" <query> "}" <Atom> "{" <literals> "}"
<beliefs> ::=     ( <belief> )*
<belief> ::=      <ground_atom> "." | <atom> ":" - <literals> "."
<goals> ::=       <goal> ( "," <goal> )*
<goal> ::=        <ground_atom> ( "and" <ground_atom> )*
<plans> ::=       <plan> ( "," <plan> )*
<plan> ::=        <basicaction> | <composedplan>
<basicaction> ::= "ε" | <Atom> | "Send("<iv>,<iv>,<atom>")" |
                  "Java("<ident>,<atom>,<var>")" | <wff>"?" | <atom>
<composedplan> ::= "if" <wff> "then" <plan> ( "else" <plan> )? |
                  "while" <query> "do" <plan> |
                  <plan> ";" <plan>
<p_rules> ::=     <p_rule> ( "," <p_rule> )*
<p_rule> ::=      <atom> "<->" <query> "|" <plan>
<p_rule> ::=      "<->" <query> "|" <plan>
<r_rules> ::=     <r_rule> ( "," <r_rule> )*
<r_rule> ::=      <plan> "<->" <query> "|" <plan>
<literals> ::=   <literal> ( "," <literal> )*
<literal> ::=    <atom> | "not("<atom>")"
<wff> ::=        <literal> | <wff> "and" <wff> | <wff> "or" <wff>
<query> ::=      <wff> | "true"
<iv> ::=         <ident> | <var>

```